In[1]:=
```
SetDirectory["~/KappaLib/"];
<< kappaLib-1.2.m
<< helper.m
```

Loading KappaLib v1.2

Loading helper.m..

- **Every medium with a double light cone can not be written as**

  kappa = C1 ast_g + rho overline(A) otimes A + C2 Id

  where A is simple.

- **Define counter example.**

  Intuition: From the proof of our main result, we see that bivector A is not simple when a1 != a2 in Metaclass I.

In[4]:=
$$\texttt{kappa = emMatrixToKappa}\left[\begin{pmatrix} a1 & 0 & 0 & -b1 & 0 & 0 \\ 0 & a2 & 0 & 0 & -b2 & 0 \\ 0 & 0 & a3 & 0 & 0 & -b3 \\ b1 & 0 & 0 & a1 & 0 & 0 \\ 0 & b2 & 0 & 0 & a2 & 0 \\ 0 & 0 & b3 & 0 & 0 & a3 \end{pmatrix}\right];$$

```
kappa = kappa //. {b3 → b2, a3 → a2};
kappa = kappa //. {a1 → 1, a2 → 2, b1 → 1, b2 → 1};
emKappaToMatrix[kappa] // MatrixForm
```

Out[7]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & -1 & 0 & 0 \\ 0 & 2 & 0 & 0 & -1 & 0 \\ 0 & 0 & 2 & 0 & 0 & -1 \\ 1 & 0 & 0 & 1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 2 & 0 \\ 0 & 0 & 1 & 0 & 0 & 2 \end{pmatrix}$$

- **Verify that kappa is invertible, skewon-free and has a double light cone**

In[8]:=
```
emDet[kappa]
Union[Flatten[emPoincare[kappa] - kappa]]
```

Out[8]= 50

Out[9]= {0}

In[10]:=
```
vars = {x0, x1, x2, x3};
fresnel = emKappaToFresnel[kappa, vars];
```

In[12]:=
$$\texttt{AA} = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & \frac{1}{2}\left(-3+\sqrt{5}\right) & 0 \\ 0 & 0 & 0 & \frac{1}{2}\left(-3+\sqrt{5}\right) \end{pmatrix};$$

$$\texttt{BB} = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & \frac{1}{2}\left(-3-\sqrt{5}\right) & 0 \\ 0 & 0 & 0 & \frac{1}{2}\left(-3-\sqrt{5}\right) \end{pmatrix};$$

In[14]:=
```
Sort[Eigenvalues[AA] // N]
Sort[Eigenvalues[BB] // N]
```

Out[14]= {-1., -0.381966, -0.381966, 1.}

Out[15]= {-2.61803, -2.61803, -1., 1.}

In[16]:=
```
Simplify[fresnel - (vars.AA.vars) (vars.BB.vars)]
```

Out[16]= 0

- **If A ∧ A = 0, then A is simple, and there are vectors u and v such that**

  A = u ∧ v

In[17]:=
```
u = {u1, u2, u3, u4};
v = {v1, v2, v3, v4};
Abivector = Table[
   u[[i]] v[[j]] - u[[j]] v[[i]],
   {i, 1, 4}, {j, 1, 4}];
```

In[20]:=
```
Abivector // MatrixForm
```

Out[20]//MatrixForm=

0	- u2 v1 + u1 v2	- u3 v1 + u1 v3	- u4 v1 + u1 v4
u2 v1 - u1 v2	0	- u3 v2 + u2 v3	- u4 v2 + u2 v4
u3 v1 - u1 v3	u3 v2 - u2 v3	0	- u4 v3 + u3 v4
u4 v1 - u1 v4	u4 v2 - u2 v4	u4 v3 - u3 v4	0

- **Define coefficients for g^ij**

In[21]:=
```
InvMetric = emMatrix["g", 4, Structure → "Symmetric"];
```

- **Write representation for kappa:**

In[22]:=
```
kappaAlt = C1 emQMedium[SqrtAbsDetG, InvMetric] +
   emBiProduct[rho, Abivector, Abivector] + C2 emIdentityKappa[];
eqs = Union[Flatten[FullSimplify[kappa - kappaAlt]]];
eqs = simp[eqs];
```

In[25]:=
```
show[eqs]
```

Out[25]//MatrixForm=

```
1  : 0
2  : 1 + C1 (g12^2 - g11 g22) SqrtAbsDetG - 2 rho (u2 v1 - u1 v2)^2
3  : 1 + C1 (g14^2 - g11 g44) SqrtAbsDetG - 2 rho (u4 v1 - u1 v4)^2
4  : - 1 + C1 (g23^2 - g22 g33) SqrtAbsDetG - 2 rho (u3 v2 - u2 v3)^2
5  : 1 + C1 (- g24^2 + g22 g44) SqrtAbsDetG + 2 rho (u4 v2 - u2 v4)^2
6  : - 1 + C1 (g34^2 - g33 g44) SqrtAbsDetG - 2 rho (u4 v3 - u3 v4)^2
7  : - 1 + C1 (- g13^2 + g11 g33) SqrtAbsDetG + 2 rho (u3 v1 - u1 v3)^2
8  : C1 (g12 g13 - g11 g23) SqrtAbsDetG - 2 rho (u2 v1 - u1 v2) (u3 v1 - u1 v3)
9  : C1 (g13 g23 - g12 g33) SqrtAbsDetG - 2 rho (u3 v1 - u1 v3) (u3 v2 - u2 v3)
10 : C1 (g12 g14 - g11 g24) SqrtAbsDetG - 2 rho (u2 v1 - u1 v2) (u4 v1 - u1 v4)
11 : C1 (g13 g14 - g11 g34) SqrtAbsDetG - 2 rho (u3 v1 - u1 v3) (u4 v1 - u1 v4)
12 : C1 (g23 g24 - g22 g34) SqrtAbsDetG - 2 rho (u3 v2 - u2 v3) (u4 v2 - u2 v4)
13 : C1 (g14 g24 - g12 g44) SqrtAbsDetG - 2 rho (u4 v1 - u1 v4) (u4 v2 - u2 v4)
14 : C1 (g14 g34 - g13 g44) SqrtAbsDetG - 2 rho (u4 v1 - u1 v4) (u4 v3 - u3 v4)
15 : C1 (g24 g34 - g23 g44) SqrtAbsDetG - 2 rho (u4 v2 - u2 v4) (u4 v3 - u3 v4)
16 : C1 (- g12 g13 + g11 g23) SqrtAbsDetG + 2 rho (u2 v1 - u1 v2) (u3 v1 - u1 v3)
17 : C1 (- g13 g23 + g12 g33) SqrtAbsDetG + 2 rho (u3 v1 - u1 v3) (u3 v2 - u2 v3)
18 : C1 (g13 g22 - g12 g23) SqrtAbsDetG + 2 rho (u2 v1 - u1 v2) (- u3 v2 + u2 v3)
19 : C1 (- g13 g14 + g11 g34) SqrtAbsDetG + 2 rho (u3 v1 - u1 v3) (u4 v1 - u1 v4)
20 : C1 (- g23 g24 + g22 g34) SqrtAbsDetG + 2 rho (u3 v2 - u2 v3) (u4 v2 - u2 v4)
21 : C1 (- g14 g24 + g12 g44) SqrtAbsDetG + 2 rho (u4 v1 - u1 v4) (u4 v2 - u2 v4)
22 : C1 (g14 g22 - g12 g24) SqrtAbsDetG + 2 rho (u2 v1 - u1 v2) (- u4 v2 + u2 v4)
23 : C1 (- g24 g34 + g23 g44) SqrtAbsDetG + 2 rho (u4 v2 - u2 v4) (u4 v3 - u3 v4)
24 : C1 (g14 g33 - g13 g34) SqrtAbsDetG + 2 rho (u3 v1 - u1 v3) (- u4 v3 + u3 v4)
25 : C1 (g24 g33 - g23 g34) SqrtAbsDetG + 2 rho (u3 v2 - u2 v3) (- u4 v3 + u3 v4)
26 : C1 (- g14 g22 + g12 g24) SqrtAbsDetG - 2 rho (u2 v1 - u1 v2) (- u4 v2 + u2 v4)
27 : C1 (- g14 g33 + g13 g34) SqrtAbsDetG - 2 rho (u3 v1 - u1 v3) (- u4 v3 + u3 v4)
28 : 2 - C2 + C1 (g13 g24 - g12 g34) SqrtAbsDetG - 2 rho (u3 v2 - u2 v3) (u4 v1 - u1 v4)
29 : 1 - C2 + C1 (g14 g23 - g13 g24) SqrtAbsDetG - 2 rho (u2 v1 - u1 v2) (u4 v3 - u3 v4)
30 : - C2 + C1 (- g14 g23 + g12 g34) SqrtAbsDetG + 2 (1 + rho (u3 v1 - u1 v3) (u4 v2 - u2 v4))
```

In[26]:=
```
gb = GroebnerBasis[eqs, Variables[eqs]]; // Timing
```

Out[26]=
```
{36.9784, Null}
```

In[27]:=
```
gb = simp[gb]; // Timing
```

Out[27]=
```
{0.000055, Null}
```

In[28]:=
```
gb
```

Out[28]=
```
{1}
```

- **We know that a system of polynomial equations has no solution in the complex domain if a Gröbner basis for the equations is {1}.**